
Introduction to Computer Architecture exam (1 hour)

Renaud Pacalet - 2026-01-26

The text in black is the original one.

The text in red is examples of the expected correct answers. Only this text was expected, 
possibly in shorter form, nothing more.

The text in blue is extra comments about the expected correct answers.

Warning: the course changes frequently (content, vocabulary, examples…); some questions 
and answer proposals can thus be partly or completely out of scope. Warning: some questions 
can be answered in many different ways; the proposed answers are just examples and they 
are not exhaustive.

The exam is closed book, closed notes, except one A4-sized, double sided, cheat 
sheet. Calculators are allowed but communicating devices are strictly forbidden. 
Please number the different pages of your paper and indicate on each page your first and 
last names. You can write your answers in French or in English, as you wish. Precede your 
answers with the question’s number. If some information or hypotheses are missing to answer 
a question, add them. If you consider a question as absurd and thus decide to not answer, 
explain why. If you do not have time to answer a question but know how to, briefly explain 
your ideas. Advice: quickly go through the document and answer the easy parts first.

The first and second questions are worth 6 points each. The third question is worth 8 points.

1. CMOS logic
The qux logic gate has three inputs A, B, C, and one output X. Its CMOS schematic is shown 
on Figure 1.

Figure 1:  The qux logic gate

1. Write its truth table.

The truth table of qux is shown in Table 1.

A B C X

0 0 0 1

0 0 1 1

0 1 0 1

0 1 1 1

1 0 0 1

1 0 1 0

1 1 0 0

1 1 1 0

Table 1: Truth table

2. Write the boolean equation of the X output of qux using the NOT, AND and OR operators 
and parentheses. Do not assume any precedence between the boolean operators, use 
parentheses to make your equation non ambiguous: NOT A AND B is ambiguous; (NOT A) 
AND B is not.

From the truth table we can deduce the boolean equation: 𝑋 =
NOT((𝐴 AND (𝐵 OR 𝐶))

3. Can you find another CMOS schematic for qux with less transistors? If not explain why, 
else draw the optimized CMOS schematic.

An optimized schematic for qux with 6 transistors instead of 8 is shown on Figure 2.

Figure 2:  The optimized qux logic gate

4. Imagine a graphical symbol for qux and draw it.

A possible symbol for qux could be as shown on Figure 3.

Figure 3:  Symbol for the qux logic gate

2. Binary representation of numbers
Notations:

• Unless otherwise stated, numeric values are in base 10.
• 𝑋mod𝑌 , where 𝑌  is not null, is the non-negative remainder of the integer division of 𝑋 

by 𝑌 : 10mod 4 = 2,−11mod 4 = 1.
• ⌊𝑉 ⌋ is the largest integral value not greater than 𝑉 : ⌊73⌋ = 2.

Let 𝑎8𝑎7𝑎6…𝑎1𝑎0 be the 9-bits two’s complement representation of signed integer 𝐴.

1. What is the range of integer values that 𝐴 can take?

𝐴 ∈ {−256,…, 255}

2. What is the 9-bits two’s complement representation of 𝐴mod 128?

00𝑎6𝑎5…𝑎1𝑎0

3. Let 𝑎8𝑎7𝑎6…𝑎1𝑎0111 be the 12-bits two’s complement representation of integer 𝐵. What 
is the mathematical relation between integers 𝐴 and 𝐵?

𝐵 = 8𝐴 + 7

4. What is the 9-bits two’s complement representation of ⌊ 𝐴
16⌋?

𝑎8𝑎8𝑎8𝑎8𝑎8𝑎7𝑎6𝑎5𝑎4

5. What is the 9-bits two’s complement representation of 202?

011001010

6. What is the 9-bits two’s complement representation of −162?

101011110

3. RISC-V assembly
In this question we use RV32I, the RISC-V Instruction Set Architecture (ISA) without 
multiplications and divisions, and the ILP32 Application Binary Interface (ABI) seen during 
lectures and labs. Reminder: according the ILP32 ABI, the size of a stack frame must be a 
multiple of 16 bytes; the general purpose saved registers are sp, gp, tp, s0, s1, …, s11; the 
general purpose non-saved registers are ra, t0, t1, …, t6, a0, a1, …, a7. Use the provided 
RISC-V cheat sheet if you don’t remember the RV32I ISA or the ILP32 ABI.

Important: only basic instructions are allowed, pseudo-instructions are forbidden.

1. Assembly coding of function foo

Write in RV32I assembly language a function named foo which input parameters 𝐴,𝐵 
are two 32-bits unsigned numbers, and that returns |𝐴 − 𝐵|, the absolute value of their 

difference. Your code must fully comply with the ILP32 ABI. The returned value must 
be always correct (no overflows). Comment each instruction on the same line after a # 
sign as in the following example:

1 addi t0,t1,0   # t0 = t1 + 0 = t1

Listing 1 shows a possible source code of foo.

1 .text
2 foo: # inputs A,B in a0,a1, output in a0
3   bgeu a0,a1,foo1  # if a0>=a1 (A>=B) goto foo1; unsigned comparison
4   sub  a0,a1,a0    # a0 = a1-a0 (B-A)
5   jalr zero,0(ra)  # return
6 foo1:
7   sub  a0,a0,a1    # a0 = a0-a1 (A-B)
8   jalr zero,0(ra)  # return

Listing 1: The foo function

Per the ILP32 ABI, function inputs are passed in registers a0, a1, …, a7 and results 
are returned in registers a0 and a1. So, foo receives its two inputs 𝐴,𝐵 in a0, a1, and 
must return its final result in a0. We first test if 𝐴 ≥ 𝐵 (as 𝐴 and 𝐵 are unsigned 
we use the bgeu instruction, not bge that compares signed numbers); if it is, we go to 
foo1, compute a0 = a0-a1 = 𝐴−𝐵 and return. Else we compute a0 = a1-a0 = 𝐵 −𝐴 
and return. There is no need for a stack frame because we do not call other functions 
and we do not modify any saved register.

Note: we cannot subtract first, test the sign of the result, and take the negation if it 
is, as in:

sub a0,a0,a1

blt a0,zero,negative

...

negative:

sub a0,zero,a0

because the inputs are unsigned numbers; if 𝐴 > 231 and 𝐵 = 0, for example, 𝐴−𝐵 >
231 would be considered negative while it is not.

2. Assembly coding of function bar

Assume function foo has been coded not by you but by one of your colleagues; you do not 
know anything about its code, except that it fully complies with the ILP32 ABI.

Write in RV32I assembly language a function named bar which input parameters 
𝐴,𝐵,𝐶,𝐷 are four 32-bits unsigned numbers, and that:

• calls function foo to compute 𝑋 = |𝐴 − 𝐵|,
• calls function foo to compute 𝑌 = |𝐶 −𝐷|,
• calls function foo to compute 𝑍 = |𝑋 − 𝑌 |,
• computes 𝑇 = ⌊𝑋+𝑌

2 ⌋ where ⌊𝑉 ⌋ is the largest integral value not greater than 𝑉 ,
• returns 𝑍 and 𝑇 .

Your code must also fully comply with the ILP32 ABI. The returned values must be 
always correct (no overflows). Comment each instruction on the same line after a # sign.

Listing 2 shows a possible source code of bar.

1 .text
2 bar: # inputs A,B,C,D in a0,a1,a2,a3, outputs Z,T in a0,a1
3   addi sp,sp,-16   # allocate 16 bytes stack frame 
4   sw   ra,0(sp)    # save ra in stack frame 
5   sw   a2,4(sp)    # save a2 (C) in stack frame 
6   sw   a3,8(sp)    # save a3 (D) in stack frame 
7   jal  ra,foo      # a0 = foo(a0,a1) (|A-B| = X) 
8   sw   a0,12(sp)   # save a0 (X) in stack frame 
9   lw   a0,4(sp)    # restore C from stack frame in a0 
10   lw   a1,8(sp)    # restore D from stack frame in a1 
11   jal  ra,foo      # a0 = foo(a0,a1) (|C-D| = Y) 
12   lw   a1,12(sp)   # restore X from stack frame in a1 
13   and  t0,a0,a1    # t0 = a0 and a1 (Y and X) 
14   andi t0,t0,1     # t0 = t0 and 1 (U = Y and X and 1) 
15   srli t1,a0,1     # t1 = a0 >> 1 (Y/2) 
16   srli t2,a1,1     # t2 = a1 >> 1 (X/2) 
17   add  t1,t1,t2    # t1 = t1 + t2 (Y/2 + X/2) 
18   add  t1,t1,t0    # t1 = t1 + t0 (T = Y/2 + X/2 + U) 
19   sw   t1,4(sp)    # save t1 (T) in stack frame 
20   jal  ra,foo      # a0 = foo(a0,a1) (|Y-X| = |X-Y| = Z) 
21   lw   a1,4(sp)    # restore T from stack frame in a1 
22   lw   ra,0(sp)    # restore ra from stack frame 
23   addi sp,sp,16    # restore sp 
24   jalr zero,0(ra)  # return 

Listing 2: The bar function

Per the ILP32 ABI, function inputs are passed in registers a0, a1, …, a7 and results 
are returned in registers a0 and a1. So, function bar receives its inputs 𝐴,𝐵,𝐶,𝐷 in 
a0, a1, a2, a3, and must return its final results 𝑍, 𝑇  in a0, a1. As bar calls another 
function (foo) it must allocate a stack frame (Line 3), at least to store its own return 
address that it receives in ra (Line 4). It also uses the stack frame to store values in 
non-saved registers that must be preserved across function calls. 16 bytes are enough.

bar saves a2, a3 in its stack frame (Lines 5 and 6) before the first call of foo because 
foo could modify these non-saved registers. It then calls foo (Line 7) to compute 𝑋 =
|𝐴 − 𝐵| (𝐴 is already in a0 and 𝐵 in a1); the result is returned in a0 and immediately 
saved in the stack frame (Line 8). Next, 𝐶 and 𝐷 are restored from the stack frame, 
respectively in a0 and a1 (Lines 9 and 10), and foo is called again (Line 11) to compute 
𝑌 = |𝐶 −𝐷| (again returned in a0). 𝑋 is then restored from the stack frame in a1 
(Line 12).

What comes next is a way to compute 𝑇 = ⌊𝑋+𝑌
2 ⌋ without overflows. It is based on 

the fact that ⌊𝑋+𝑌
2 ⌋ = ⌊𝑋2 ⌋ + ⌊𝑌2 ⌋ + 𝑈  where 𝑈 = 1 if 𝑋 and 𝑌  are odd, else 𝑈 =

0. As ⌊𝑋2 ⌋, ⌊
𝑌
2 ⌋ ∈ {0,…, 231 − 1} there cannot be an overflow when computing their 

sum. So, 𝑈  is computed in t0 with bitwise AND operations (Lines 13 and 14), 𝑌  and 
𝑋 are divided by 2 with logical right shifts (Lines 15 and 16), and 𝑇 = ⌊𝑋2 ⌋ + ⌊𝑌2 ⌋ +
𝑈  is computed in t1 with regular add instructions (Lines 17 and 18). t1 (containing 
𝑇 ) is then saved in the stack frame (Line 19) because there is one more call to foo to 
come that could overwrite this non-saved register.

Finally, foo is called one last time with 𝑌  still in a0 and 𝑋 in a1, returning 𝑍 = |𝑋 −
𝑌 | in a0 (Line 20). 𝑇  is restored from the stack frame in a1 (Line 21); with 𝑍 in a0 
and 𝑇  in a1 the function is almost ready to return. Before that, it must only restore 
ra from the stack frame (Line 22), restore sp by adding 16 to it (Line 23), and return 
(Line 24).

